


Circuit rezonant – LC paralel

Scopul lucrarii	1
Descrierea circuitului	1
Ecuatii de stare	1
Ecuatii TTN	2
Calculul functiei de transfer H(s).....	2
Metoda I: divizor de tensiune	2
Metoda II: ecuatii TTN	3
Metoda III: folosind calcul simbolic	3
Functia de transfer in regim permanent H(j ω)	4
Diagrame frecventiale	6
Forma functiei de transfer la variatia pozitiei polilor.....	7
Raspuns de regim permanent	7
Raspunsul la semnal armonic	7
Raspunsul la semnal armonic de frecventa de rezonanta a filtrului	8
Comportarea circuitului in jurul frecventei de rezonanta.....	8
Raspuns de regim tranzitoriu.....	9
Raspunsul la semnal treapta (functia pondere)	10
Modificarea functiei pondere la variatia pozitiei polilor.....	10
Simulare SPICE a circuitului	11
Diagrama Bode de modul si faza	12
Raspuns tranzitoriu.....	12
Raspuns permanent	13

Scopul lucrarii

In lucrarea de fata ne propunem analiza unui circuit liniar de tip Trece Banda (TB) de ordin II cu grupare paralel bobina si condensator. Schema circuitului este:



Descrierea circuitului

> **restart:with(Syrup):**

Descrierea circuitului folosind un netlist de tip spice:

```
> FTB :=  
"filtru trece banda  
Vg 1 0  
R 1 2  
L 2 0  
C 2 0  
.end":
```

Ecuatii de stare

```
> restart:with(Syrup):FTB := "filtru trece banda\nVg 1 0\nR 1  
2\nL 2 0\nC 2 0\n.end":
```

Descrierea folosind ecuatii de stare a circuitului

1) **ecuatia de stare (vectorial):**

> **syryp(FTB, tran, 'curenti','tensiuni');**

$$\{v_C(0)=0, i_L(0)=0, \frac{\partial}{\partial t} v_C(t) = -\frac{i_L(t)R - Vg + v_C(t)}{CR}, \frac{\partial}{\partial t} i_L(t) = \frac{v_C(t)}{L}\},$$

$$\{v_C(t), i_L(t)\}$$

2) **ecuatii de iesire** se pot scrie in functie de tensiunile de noduri si curentii prin laturi:

> **tensiuni;**

$$\{v_1 = Vg, v_2 = v_C(t)\}$$

> **curenti;**

$$\{i_L = i_L(t), i_{Vg} = -\frac{Vg - v_C(t)}{R}, i_R = \frac{Vg - v_C(t)}{R}, i_C = -\frac{i_L(t)R - Vg + v_C(t)}{R}\}$$

Obs: Ordinul circuituluie este egal cu 2. Descrierea functionarii circuitului se face cu un sistem de 2 ecuatii diferentiale de ordin I. Variabilele de stare sunt: tensiunea pe condensator si curentul prin bobina. In functie de aceste variabile se exprima curentii si tensiunile din circuit.

Ecuatii TTN

> **restart:with(Syrup):FTB := "filtru trece banda\nVg 1 0\nR 1 2\nL 2 0\nC 2 0\n.end":**

Descrierea circuitului folosind Teorema Tensiunilor Nodale:

> **syryp(FTB, ac, 'curenti','tensiuni');**

> **tensiuni;**

$$\{v_2 = \frac{sL Vg}{s^2 CLR + R + sL}, v_1 = Vg\}$$

> **curenti;**

$$\left\{ \begin{aligned} i_L &= \frac{Vg}{s^2 CLR + R + sL}, i_{Vg} = -\frac{Vg(s^2 CL + 1)}{s^2 CLR + R + sL}, i_R = \frac{Vg - \frac{sL Vg}{s^2 CLR + R + sL}}{R}, \\ i_C &= \frac{s^2 CL Vg}{s^2 CLR + R + sL} \end{aligned} \right\}$$

Calculul functiei de transfer H(s)

Functia de transfer a unui sistem linar se poate calcula prin mai multe metode:

Metoda I: divizor de tensiune

Calculul functiei de transfer folosind divizor de tensiune:

$$H(s) = \frac{sL \parallel \frac{1}{sC}}{R + \left(sL \parallel \frac{1}{sC}\right)} = \frac{s \frac{1}{RC}}{s^2 + s \frac{1}{RC} + \frac{1}{LC}}$$

Notam:

$$2a = \frac{1}{RC}, \quad \omega_0^2 = \frac{1}{LC}$$

Deci functia de transfer devine:

$$H(s) = \frac{2as}{s^2 + 2as + \omega_0^2}$$

Se defineste factorul de calitate :

$$Q = \frac{\omega_0}{2a}$$

In cazul nostru factorul de calitate este:

$$Q = R\sqrt{\frac{C}{L}}$$

Se calculeaza polii si zerourile functiei de transfer.

zerou

$$z = 0$$

poli:

$$p_1 = -a - \sqrt{a^2 - \omega_0^2}$$

$$p_2 = -a + \sqrt{a^2 - \omega_0^2}$$

In functie de factorul de calitate Q avem doua cazuri:

Cazul I ($Q < \frac{1}{2}$) avem doi poli reali distincti.

Cazul II: ($Q = \frac{1}{2}$) avem un pol dublu.

Cazul III ($Q > \frac{1}{2}$) avem poli complecsi conjugati.

Metoda II: ecuatii TTN

Pentru circuitul cu nodurile din figura se scrie TTN:

$$\begin{cases} V_{10}(s) = Vg(s) \\ -GV_{10}(s) + \left(G + \frac{1}{sL} + sC\right)V_{20}(s) = 0 \end{cases}$$

In urma calculelor obtinem aceeasi functie de transfer ca si prin metoda I:

$$H(s) = \frac{2as}{s^2 + 2as + \omega_0^2}$$

Metoda III: folosind calcul simbolic

Folosind Maple si pachetul Syrup se descrie circuitul

```
> restart:with(Syrup):FTB := "filtru trece sus\nVg 1 0\nR 1 2\nL
2 0\nC 2 0\n.end":
> libname:="c://maple//SCSlib",libname:
> syrup(FTB, ac, 'curenti','tensiuni'):
> Hs:=eval(v[2]/v[1],tensiuni);
```

$$Hs := \frac{sL}{s^2 CLR + R + sL}$$

```
> Hs:=sort(simplify(eval(Hs,[C=1/(2*R*alpha),
L=(2*R*alpha)/(omega0^2)])),s);
```

$$Hs := 2 \frac{\alpha s}{s^2 + 2 \alpha s + \omega 0^2}$$

Zerourile functiei de transfer:

```
> z:=solve(numer(Hs)=0,s);
```

$$z := 0$$

Polii functiei de transfer:

```
> p:={solve(denom(Hs)=0,s)};
```

$$p := \{-\alpha + \sqrt{\alpha^2 - \omega 0^2}, -\alpha - \sqrt{\alpha^2 - \omega 0^2}\}$$

Evaluare numerica:

```
> alpha:=1/(2*R*C):omega0:=1/sqrt(L*C):
```


```
> H:=eval(Hs,[L=0.002, C=22*1E-9, R=1000]):
```

Polii si zerourile functiei de transfer:

```
> PZ[numeric](H,s);
```

$$\begin{bmatrix} z1 & 0. \\ p1 & -22720. + 149000. I \\ p2 & -22720. - 149000. I \end{bmatrix}$$

```
> PZ[grafic](H,s);
```




Interpretarea functiei de transfer:

```
> plot3d(abs(eval(H,s=sigma+I*omega)),sigma=-50000..50000,omega=-300000..300000,numpoints=2500,axes=normal,title="Reprezentarea in spatiu a modulului F.D.T.");
```

```
plot3d(argument(eval(H,s=sigma+I*omega)),sigma=-50000..50000,omega=-300000..300000,numpoints = 2500,axes=normal,title="Reprezentarea in spatiu pentru argumentul F.D.T.");
```

Reprezentarea in spatiu a modulului F.D.T.



Funcția de transfer in regim permanent H(j?)

```
> restart:with(Syrup):
```

```
> libname:="../SCSlib","../DCElib",libname:
```

```
> Hs := (2*alpha*s) / (s^2 + 2*alpha*s + omega0^2);
```

$$Hs := 2 \frac{\alpha s}{s^2 + 2 \alpha s + \omega_0^2}$$

```
> Homega := subs(s=I*omega, Hs);
```

$$Homega := \frac{2 I \alpha \omega}{-\omega^2 + 2 I \alpha \omega + \omega_0^2}$$

Valori ale functiei de transfer $H(j\omega)$ pentru $\omega = \{0, \omega_0, \infty\}$:

```
> H0 := limit(Homega, omega=0); Hinf := limit(Homega, omega=infinity);
```

```
abs_Homega0 := abs(simplify(eval(Homega, omega=omega0)));
```

```
arg_Homega0 := argument(simplify(eval(Homega, omega=omega0)));
```

$$H0 := 0$$

$$Hinf := 0$$

$$abs_Homega0 := 1, \quad arg_Homega0 := 0$$

```
> alpha := 1 / (2*R*C); omega0 := 1 / sqrt(L*C);
```

Pulsatia de rezonanta:


```
> val_omega0 := eval(omega0, [L=0.002, C=22*1E-9, R=1000]);
```

$$val_omega0 := 150755.6723$$


```
> plot(abs(eval(Homega, [L=0.002, C=22*1E-9, R=1000])), omega=-400000..400000, numpoints=200, axes=normal, title="Reprezentarea  
modulului F.D.T. pentru regim permanent");
```

```
plot(argument(eval(Homega, [L=0.002, C=22*1E-9, R=1000])), omega=-400000..400000, numpoints=200, axes=normal, title="Reprezentarea  
argumentului F.D.T. pentru regim permanent");
```

Reprezentarea modulului F.D.T. pentru regim permanent




Reprezentarea argumentului F.D.T. pentru regim permanent



```
> plots[spacecurve]([omega, Re(eval(Homega, [L=0.002, C=22*1E-9, R=1000])), Im(eval(Homega, [L=0.002, C=22*1E-9, R=1000])), omega=-600000..600000], numpoints = 10000, axes=normal,
```

title="Reprezentarea spatiala a F.D.T. pentru regim permanent");

Reprezentarea spatiala a F.D.T. pentru regim permanent



Diagrame frecventiale

> restart; libname:="c://maple//SCSlib", "../DCElib", libname:

> Hs:=(2*alpha*s)/(s^2+2*alpha*s+omega0^2);

$$Hs := 2 \frac{\alpha s}{s^2 + 2 \alpha s + \omega_0^2}$$

> alpha:=1/(2*R*C):omega0:=1/sqrt(L*C):

> H:=eval(Hs,[L=0.002, C=22*1E-9, R=1000]):

> cat("omega0 = ",convert(eval(omega0,[L=0.002, C=22*1E-9, R=1000]),string)," rad/s");

"omega0 = 150755.6723 rad/s"

> Bode[castig](H,numarpuncte=400);Bode[faza](H,numarpuncte=400);

Diagrama Bode de castig





Diagrama Bode de faza




>cat("H(10^4)=",convert(20*log10(abs(eval(H,s=I*10^4))),string)," dB"); cat("arg(H(10^4))=",convert(argument(eval(H,s=I*10^4)),string)," grade");

"H(10^4)=-33.94285006dB"

"arg(H(10^4))=1.550710639grade"

> **Bode[polara](H,numarpuncte=600);**



Forma functiei de transfer la variatia pozitiei polilor

Pozitia polilor se modifica la modificarea valorii rezistentei R. Valoarea frecventei de rezonanta ramine constanta.

> **z:=solve(numer(eval(subs(s=I*omega,Hs),[L=0.002,C=22*1E-9]))=0,omega):**

> **p:=solve(denom(eval(subs(s=I*omega,Hs),[L=0.002,C=22*1E-9]))=0,omega):**

> **plots[animate]({[(evalf(Im(p[1])),evalf(Re(p[1])))],[(evalf(Im(p[2])),evalf(Re(p[2])))]},R=50..800,frames=750,style=point):**

plots[animate]([log10(omega),20*log10(abs(eval(subs(s=I*omega,Hs),[L=0.002,C=22*1E-9]))),omega=10^4..10^7],R=50..2000,frames=75,numpoints=750,title="Diagrama de cistig"):

plots[animate]([log10(omega),argument(eval(subs(s=I*omega,Hs),[L=0.002,C=22*1E-9])),omega=10^4..10^7],R=50..2000,frames=75,numpoints=750,title="Diagrama de faza"):

Raspuns de regim permanent

Pentru calculul raspunsului permanent se prefera efectuarea calculelor in domeniul frecventa (din motive de simplitate – convolutia din domeniul timp se inlocuieste cu produs in domeniul frecventa)

$$1) e(t) \rightarrow E(\mathbf{w}) = F\{e(t)\}$$

$$2) Y(\mathbf{w}) = E(\mathbf{w})H(\mathbf{w})$$

$$3) y(t) = F^{-1}\{Y(\mathbf{w})\}$$

> **restart:libname:="c://maple//SCSlib",libname:**

> **F:=table([dir=inttrans[fourier],inv=inttrans[invfourier]]):**

> **Hs:=(2*alpha*s)/(s^2+2*alpha*s+omega0^2);**

$$Hs := 2 \frac{\alpha s}{s^2 + 2 \alpha s + \omega 0^2}$$

> **Homega:=subs(s=I*omega,Hs);**

$$Homega := \frac{2 I \alpha \omega}{-\omega^2 + 2 I \alpha \omega + \omega 0^2}$$

Raspunsul la semnal armonic

In acest caz expresia excitatiei $e(t)$ este de forma:

> **e:=A0*cos(w*t);**

$$e := A0 \cos(\omega t)$$

Transformata Fourier a excitatie $e(t)$ este:

> **E:=F[dir](e,t,omega);**

$$E := A0 (\pi \text{Dirac}(-\omega + \omega) + \pi \text{Dirac}(\omega + \omega))$$

Transformata Fourier a excitatie $y(t)$ este:

> **Y:=H*omega*E;**

$$Y := \frac{2 I \alpha \omega A0 (\pi \text{Dirac}(-\omega + \omega) + \pi \text{Dirac}(\omega + \omega))}{-\omega^2 + 2 I \alpha \omega + \omega_0^2}$$

Raspunsul $y(t)$ al circuitului la excitatia $e(t)$ este:

> **y:=simplify(normal(convert(F[inv](Y,omega,t),trig),expanded));**

$$y := 2 \frac{\alpha A0 \omega (w^2 \sin(\omega t) + 2 \alpha \omega \cos(\omega t) - \omega_0^2 \sin(\omega t))}{w^4 - 2 w^2 \omega_0^2 + 4 \alpha^2 \omega^2 + \omega_0^4}$$

Obs: Formula de mai sus poate fi obtinuta si astfel. Se porneste de la functia de transfer $H(j\omega)$:

$$H(j\omega) = \frac{2a j\omega}{-\omega^2 + 2a j\omega + \omega_0^2}$$

Se calculeaza modulul si argumentul:

$$\begin{cases} |H(j\omega)| = \frac{2a\omega}{\sqrt{(\omega_0^2 - \omega^2)^2 + (2a\omega)^2}} \\ \arg(H(j\omega)) = \frac{\pi}{2} - \text{arctg} \frac{2a\omega}{\omega_0^2 - \omega^2} \end{cases}$$

Raspunsul permanent este:

$$y(t) = A_0 |H(j\omega)| \cos(\omega t + \arg(H(j\omega)))$$

$$y(t) = A_0 \frac{2a\omega}{\sqrt{(\omega_0^2 - \omega^2)^2 + (2a\omega)^2}} \cos\left(\omega t + \frac{\pi}{2} - \text{arctg} \frac{2a\omega}{\omega_0^2 - \omega^2}\right)$$

Raspunsul la semnal armonic de frecventa de rezonanta a filtrului

Se calculeaza raspunsul circuitului la frecventa egala cu frecventa de rezonanta a filtrului:

> **e1:=eval(e,w0=omega0);**

$$e1 := A0 \cos(\omega_0 t)$$

> **y1:=simplify(eval(y,w0=omega0));**

$$y1 := A0 \cos(\omega_0 t)$$

Comportarea circuitului in jurul frecventei de rezonanta

Pentru circuitul fizic s-au facut notatiile:

> **alpha:=1/(2*R*C):omega0:=1/sqrt(L*C):**

Pulsatia de rezonanta a circuitului este:

> **omega0_val:=eval(omega0, [L=0.002, C=22*1E-9, R=1000]);**


$$\omega_{0_val} := 150755.6723$$

Figura Lisajoux la frecventa de rezonanta a circuitului devine o dreapta:

> **plot(eval([e1,y1,t=-1/omega0_val*Pi..1/omega0_val*Pi],[L=0.002, C=22*1E-9, R=1000,A0=1]),title="Figura Lissajoux la frecventa de**

taiere a filtrului");

Figura Lissajoux la frecventa de taiere a filtrului



Vom urmarii modul in care se modifica semnalul de iesire din circuit (si figura Lisajoux) la variatia frecventei semnalului de intrare. Se alege o variatie a frecventei in jurul valorii frecventei de rezonanta.

```
> wd:=omega0_val*evalf([seq(10^(i/20),i=-10..10)]):
> INTERFACE_PLOT(ANIMATE(seq([CURVES(op(op(plot(eval(e,[A0=1,w0=wd
[i]]),t=-
Pi/wd[1]..Pi/wd[1],numpoints=200))[1])[1],COLOR(RGB,1,0,0)),CURVE
S(op(op(plot(eval(1.01*y,[A0=1,L=0.002,C=22*1E-
9,R=1000,w0=wd[i]]),t=-
Pi/wd[1]..Pi/wd[1],numpoints=200))[1])[1],COLOR(RGB,0,0,0)),TEXT(
[0,0.2],cat("w0=",convert(wd[i],string)),FONT(HELVETICA,8))],i=1.
.nops(wd))),AXESLABELS("t","e(t)","y(t)"),TITLE("Forma de unda a
semnalelor excitatie (rosu) si raspuns (negru)"));
INTERFACE_PLOT(ANIMATE(seq([op(plot([eval(e,[A0=1,w0=wd[i]]),eval
(y,[A0=1,L=0.002,C=22*1E-9,R=1000,w0=wd[i]]),t=-
Pi/wd[i]..Pi/wd[i],color=black))[1],TEXT([0.2,0.2],cat("w0=",con
vert(wd[i],string)),FONT(HELVETICA,8))],i=1..nops(wd))),AXESLABEL
S("e(t)","y(t)"),TITLE("Figura Lissajoux"));
```

Forma de unda a semnalelor excitatie (rosu) si raspuns (negru)





Figura Lissajoux



Raspuns de regim tranzitoriu

Pentru calculul raspunsului tranzitoriu se prefera efectuarea calculului in domeniul variabilei complexe "s" :

- 1) $e(t) \rightarrow E(s) = L\{e(t)\}$
- 2) $Y(s) = E(s)H(s)$
- 3) $y(t) = L^{-1}\{Y(s)\}$

```
> restart: libname:="c://maple//SCSlib", libname:
```

```
> L:=table([dir=inttrans[laplace],inv=inttrans[invlaplace]]):
```

>assume(_alpha,positive):assume(_omega,positive):assume(_tau,positive):assume(_T,positive):Hs:=(2*alpha*s)/(s^2+2*alpha*s+omega0^2);

$$Hs := 2 \frac{\alpha s}{s^2 + 2 \alpha s + \omega_0^2}$$

Raspunsul la semnal treapta (functia pondere)

In acest caz expresia excitatie e(t) este de forma:

> e:=A0*Heaviside(t);

$$e := A_0 \text{Heaviside}(t)$$

• Transformata Laplace a excitatie e(t) este:

> E:=L[dir](e,t,s);

$$E := \frac{A_0}{s}$$

• Transformata Laplace a excitatie y(t) este:

> Y:=Hs*E;

$$Y := 2 \frac{\alpha A_0}{s^2 + 2 \alpha s + \omega_0^2}$$

• Raspunsul y(t) al circuitului la excitatie e(t) este:

> y:=L[inv](Y,s,t)*Heaviside(t);

$$y := 2 \alpha A_0 \left(-\frac{1}{2} \frac{e^{(-\sqrt{\alpha^2 - \omega_0^2} + \alpha)t}}{\sqrt{\alpha^2 - \omega_0^2}} + \frac{1}{2} \frac{e^{(\sqrt{\alpha^2 - \omega_0^2} - \alpha)t}}{\sqrt{\alpha^2 - \omega_0^2}} \right) \text{Heaviside}(t)$$


Modificarea functiei pondere la variatia pozitiei polilor

$$e(t) = A s(t), \quad E(s) = \frac{A}{s}, \quad Y(s) = H(s)E(s) = \frac{2a A}{s^2 + 2as + w_0^2} \quad \text{unde} \quad \begin{cases} 2a = \frac{1}{RC} \\ w_0^2 = \frac{1}{LC} \end{cases}$$

In functie de pozitia polilor distingem urmatoarele 3 cazuri:


1) $a < w_0, \Delta < 0 \Rightarrow$ poli complex conjugati $(Q > \frac{1}{2})$

$$Y(s) = \frac{2aA}{(s+a)^2 + w_r^2} \quad \text{unde } w_r = \sqrt{w_0^2 - a^2} \quad y(t) = \frac{2aA}{w_r} e^{-at} \sin(w_r t) s(t)$$



2) $\mathbf{a} = \mathbf{w}_0$, $\Delta = 0 \Rightarrow$ radacini reale egale duble ($Q = \frac{1}{2}$)


$$Y(s) = \frac{2\mathbf{a}A}{(s + \mathbf{a})^2} \quad y(t) = 2\mathbf{a}Ate^{-\mathbf{a}t}\mathbf{s}(t)$$



3) $\mathbf{a} > \mathbf{w}_0$, $\Delta > 0 \Rightarrow$ radacini reale distincte ($Q < \frac{1}{2}$)

$$Y(s) = \frac{k_1}{s + \mathbf{a}_1} + \frac{k_2}{s + \mathbf{a}_2}$$

$$y(t) = (k_1e^{-\mathbf{a}_1t} + k_2e^{-\mathbf{a}_2t})\mathbf{s}(t)$$



Simulare SPICE a circuitului

```
*Filtru Trece Banda de ordin 2
Vin 1 0 1 AC 1
R1 1 2 {R}
c1 2 0 22nF
L1 2 0 2mH
.PARAM R=10k
.STEP PARAM R LIST 10 150 1k
.PROBE
.AC DEC 100 10Hz 20MEGHZ
.END
```







Diagrama Bode de modul si faza

S-au trasat pe acelasi grafic pentru mai multe valori ale rezistentei R: R=100 (verde), R=1500 (rosu) si R=1kO (albastru).



Raspuns tranzitoriu

Pentru aceleasi valori ale rezistentei s-a calculat $h(t)$ - functia pondere a circuitului:




Raspuns permanent

Pentru circuitul cu $R=1k\Omega$ se aplica la intrare semnal sinusoidal de frecventa: $f=10\text{KHz}$ (verde), $f=22.9\text{KHz}$ (rosu) si $f=100\text{KHz}$ (albastru).

Se vizualizeaza semnalul de iesire dupa ce regimul tranzitoriu s-a stins:

```
* Filtru Trece Banda de ordin 2
Vin 1 0 1 SIN(0 1 {frecv})
R1 1 2 1k
c1 2 0 22nF
L1 2 0 2mH
.PARAM frecv=100khz
.STEP PARAM frecv LIST 10k 22.9k 100k
.PROBE
.TRAN 0.0001n 600u
.END
```




Circuit rezonant - LC serie

Scopul lucrarii.....	14
Caracterizarea circuitului.....	14
Ecuatii de stare.....	14
Ecuatii TTN	15
Calculul functiei de transfer H(s)	15
Metoda I: divizor rezistiv.....	15
MetodaII: ecuatii TTN	16
Metoda III: calcul simbolic.....	16
Diagrame frecventiale.....	17
Simulare Spice	18
Functia pondere la variatia rezistentei R	14
Diagrame Bode de modul si faza la variatia rezistentei R.....	14

Scopul lucrarii

In lucrarea de fata ne propunem analiza unui circuit liniar de tip Trece Banda (TB) de ordinul II cu grupare serie bobina si condensator. Schema circuitului este:



Caracterizarea circuitului

:Descrierea circuitului folosind un netlist de tip spice

```
> restart:with(Syrup):  
> FTB :=  
"filtru trece banda  
Vg 1 0  
L 1 2  
C 2 3  
R 3 0  
.end":
```

Ecuatii de stare

```
> restart:with(Syrup):FTB := "filtru trece banda\nVg 1 0\nL 1  
2\nC 2 3\nR 3 0\n.end":
```

Descrierea folosind ecuatii de stare a circuitului

1) **ecuatia de stare (vectorial):**

```
> syrup(FTB, tran, 'curenti','tensiuni');
```

```
Syrup/parsedeck: Analyzing SPICE deck "filtru trece banda" (ignoring this  
line)
```

$$\left\{ \frac{\partial}{\partial t} i_L(t) = -\frac{-Vg + i_L(t)R + v_C(t)}{L}, \frac{\partial}{\partial t} v_C(t) = \frac{i_L(t)}{C}, v_C(0) = 0, i_L(0) = 0 \right\},$$

$$\{i_L(t), v_C(t)\}$$

2) **ecuatii de iesire** se pot scrie in functie de tensiunile de noduri si curentii prin laturi:

> **tensiuni;**

$$\{v_3 = i_L(t)R, v_2 = i_L(t)R + v_C(t), v_1 = Vg\}$$

> **curenti;**

$$\{i_{Vg} = -i_L(t), i_C = i_L(t), i_R = i_L(t), i_L = i_L(t)\}$$

Obs: Ordinul circuitului este egal cu 2. Descrierea functionarii circuitului se face cu un sistem de 2 ecuatii diferentiale de ordin I. Variabilele de stare sunt: tensiunea pe condensator si curentul prin bobina. In functie de aceste variabile se exprima curentii si tensiunile din circuit.

Ecuatii TTN

```
> restart:with(Syrup):FTB := "filtru trece banda\nVg 1 0\nL 1
2\nC 2 3\nR 3 0\n.end":
```

Descrierea circuitului folosind Teorema Tensiunilor Nodale:

```
> syrup(FTB, ac, 'curenti','tensiuni');
```

```
Syrup/parsedeck: Analyzing SPICE deck "filtru trece banda" (ignoring this
line)
```

```
> tensiuni;
```

$$\{v_2 = \frac{Vg(sCR+1)}{s^2LC+sCR+1}, v_3 = \frac{sCVgR}{s^2LC+sCR+1}, v_1 = Vg\}$$

```
> curenti;
```

$$\left\{ i_L = \frac{sCVg}{s^2LC+sCR+1}, i_{Vg} = -\frac{sCVg}{s^2LC+sCR+1}, i_C = \frac{sCVg}{s^2LC+sCR+1}, \right.$$

$$\left. i_R = \frac{sCVg}{s^2LC+sCR+1} \right\}$$

Calculul functiei de transfer H(s)

Metoda I: divizor rezistiv

Calculul functiei de transfer folosind divizor de tensiune:

$$H(s) = \frac{sL + \frac{1}{sC}}{R + sL + \frac{1}{sC}} = \frac{s\frac{R}{L}}{s^2 + s\frac{R}{L} + \frac{1}{LC}}$$

Notam:

$$2a = \frac{R}{L}, \quad w_0^2 = \frac{1}{LC}$$

Deci functia de transfer devine:

$$H(s) = \frac{2as}{s^2 + 2as + w_0^2}$$

Se defineste factorul de calitate :

$$Q = \frac{w_0}{2a}$$

In cazul nostru factorul de calitate este:

$$Q = \frac{1}{R} \sqrt{\frac{L}{C}}$$

Se calculeaza polii si zerourile functiei de transfer.
zerou

$$z = 0$$

poli:

$$p_1 = -a - \sqrt{a^2 - \omega_0^2}$$

$$p_2 = -a + \sqrt{a^2 - \omega_0^2}$$

In functie de factorul de calitate Q avem doua cazuri:

Cazul I ($Q < \frac{1}{2}$ respectiv $R > 2\sqrt{\frac{L}{C}}$) avem doi poli reali distincti.

Cazul II: ($Q = \frac{1}{2}$ respectiv $R = 2\sqrt{\frac{L}{C}}$) avem un pol dublu.

Cazul III ($Q > \frac{1}{2}$ respectiv $R < 2\sqrt{\frac{L}{C}}$) avem poli complecsi conjugati.

Metoda II: ecuatii TTN

Pentru circuitul cu nodurile din figura se scrie TTN:

$$\begin{cases} V_{10}(s) = Vg(s) \\ -\frac{1}{sL}V_{10}(s) + \left(sC + \frac{1}{sL}\right)V_{20}(s) - sCV_{30}(s) = 0 \\ -sCV_{20}(s) + \left(sC + \frac{1}{sL}\right)V_{30}(s) = 0 \end{cases}$$

Rezolvind acest sistem ajungem la acelasi rezultat.

Metoda III: calcul simbolic

```
> restart:with(Syrup):FTB := "filtru trece banda\nVg 1 0\nL 1
2\nC 2 3\nR 3 0\n.end":
```

```
> libname:="c://maple//SCSlib",libname:
```

```
> syrup(FTB, ac, 'curenti','tensiuni'):
```

```
Syrup/parsedeck: Analyzing SPICE deck "filtru trece banda" (ignoring this
line)
```

```
> Hs:=eval(v[3]/v[1],tensiuni);
```

$$Hs := \frac{s C R}{1 + s C R + s^2 C L}$$

```
> Hs:=sort(simplify(eval(Hs,[C=(2*alpha)/(omega0^2*R),
L=R/(2*alpha)]),s);
```


$$Hs := 2 \frac{\alpha s}{s^2 + 2 \alpha s + \omega_0^2}$$

Evaluare numerica:


```
> alpha:=R/(2*L):omega0:=1/sqrt(L*C):
> H:=eval(Hs,[L=0.002, C=22*1E-9, R=700]):
Polii si zerourile functiei de transfer:
> PZ[numeric](H,s);
```


$$\begin{bmatrix} z1 & 0. \\ p1 & -86150. \\ p2 & -263900. \end{bmatrix}$$

```
> PZ[grafic](H,s);
```



```
> plot3d(abs(eval(H,s=sigma+I*omega)),sigma=-400000..400000,omega=-400000..400000,numpoints=2500,axes=normal,view=[DEFAULT, DEFAULT, 0..25],title="Reprezentarea in spatiu a modulului F.D.T.");
plot3d(argument(eval(H,s=sigma+I*omega)),sigma=-400000..400000,omega=-400000..400000,numpoints = 2500, axes=normal, title="Reprezentarea in spatiu pentru argumentul F.D.T.");
```

Reprezentarea in spatiu a modulului F.D.T.



Diagrame frecventiale

```
> restart:libname:="c://maple//SCSlib",libname:
> Hs:=(2*alpha*s)/(s^2+2*alpha*s+omega0^2);
```

$$Hs := 2 \frac{\alpha s}{s^2 + 2 \alpha s + \omega_0^2}$$

```
> alpha:=R/(2*L):omega0:=1/sqrt(L*C): cat("omega0 = ",convert(eval(omega0,[L=0.002, C=22*1E-9, R=700]),string)," rad/s");
"omega0 = 150755.6723 rad/s"
```

```
> Bode[castig](eval(Hs,[L=0.002, C=22*1E-9, R=700]),numarpuncte=400);Bode[faza](eval(Hs,[L=0.002, C=22*1E-9, R=700]),numarpuncte=400);
```

Diagrama Bode de castig






Diagrama Bode de faza




> Bode[polara](eval(Hs,[L=0.002,C=22*1E-9,R=700]),numarpuncte=400);


Diagrama polara



Simulare Spice



```
*FTB
Vg 1 0 SIN(0 1 1KHz)
L 1 2 0.0001
C 2 3 22n
R 3 0 1K
.tran 0.1m 4m 0 20u
.probe
.end
```




Funcția pondere la variația rezistenței R

```

*FTB
Vg 1 0 PULSE(0 1 1n 1n 0 600u 601m )
L 1 2 2m
C 2 3 22n
R 3 0 {R}
.PARAM R=10K
.STEP PARAM R LIST 100 603 1K
.PROBE
.TRAN 0.0001n 600u
.END

```




Diagrame Bode de modul și fază la variația rezistenței R

```


















*FTB
Vg 1 0 AC 1
L 1 2 2m
C 2 3 22n
R 3 0 {R}
.PARAM R=10K
.STEP PARAM R LIST 100 603 1K
.AC DEC 100 10Hz 20MEGhz
.probe
.end

```

Circuit resonant LC



Comportarea circuitului la variatia lui R (pozitia poli-zerouri, diagramele Bode, functia pondera)

Valori	$L=0.002, C=22*1E-9, R=100$	$L=0.002, C=22*1E-9, R=603$	$L=0.002, C=22*1E-9, R=1k$
Poli-Zerouri	<p>Diagrama poli-zerouri</p> <p>$+ p1 = -.2500e5 + j.1487e8^*$</p> <p>$+ z1 = 0.$</p> <p>$+ p2 = -.2500e5 - j.1487e8^*$</p> 	<p>Diagrama poli-zerouri</p> <p>$+ p1 = -.1508e5 + j.2107e7$</p> <p>$+ z1 = 0.$</p> 	<p>Diagrama poli-zerouri</p> <p>$+ p1 = -.4404e5$</p> <p>$+ z1 = .5043e5$</p> 
Modul	<p>Diagrama Bode de castig</p>  <p>Diagrama Bode de faza</p> 	<p>Diagrama Bode de castig</p>  <p>Diagrama Bode de faza</p> 	<p>Diagrama Bode de castig</p>  <p>Diagrama Bode de faza</p> 
Faza	<p>Diagrama Bode de castig</p>  <p>Diagrama Bode de faza</p> 	<p>Diagrama Bode de castig</p>  <p>Diagrama Bode de faza</p> 	<p>Diagrama Bode de castig</p>  <p>Diagrama Bode de faza</p> 
Rasp. Transitoriu	<p>Raspunsul la excitatie data a unui FTB de ordinul II</p> 	<p>Raspunsul la excitatie data a unui FTB de ordinul II</p> 	<p>Raspunsul la excitatie data a unui FTB de ordinul II</p> 